# $Id: PKGBUILD 144938 2011-12-11 04:46:16Z eric $
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=putty
pkgver=0.62
pkgrel=1
pkgdesc="A terminal integrated SSH/Telnet client"
arch=('i686' 'x86_64')
url="http://www.chiark.greenend.org.uk/~sgtatham/putty/"
license=('MIT')
depends=('gtk2')
source=(http://the.earth.li/~sgtatham/putty/${pkgver}/${pkgname}-${pkgver}.tar.gz
        putty-${pkgver}.tar.gz.sig::http://the.earth.li/~sgtatham/putty/latest/putty-${pkgver}.tar.gz.DSA)
sha1sums=('5898438614117ee7e3704fc3f30a3c4bf2041380'
          '128cded001647e3b47494dbfb1eeb70cbed49a7c')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}/unix"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}/unix"
  make DESTDIR="${pkgdir}" install
  install -D -m644 ../LICENCE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
